```
(* ========================================================= *)
(* Asympotics formula for A032302 (c) Vaclav Kotesovec, Jan 04 2016 *)

(* f[x] = Product[1 + 2*x^k, {k, 1, Infinity}] *)
(* substitution x = Exp[-u] *)

(* Calculating of G[u] =
 Log[f(Exp[-u])] = Sum[Log[1 + 2*Exp[-u*k]],{k, 1, Infinity}] *)
(* For Euler-MacLaurin summation formula see e.g.
    https://en.wikipedia.org/wiki/Euler%E2%80%93Maclaurin_formula *)

Integrate[Log[1 + 2 * Exp[-u * x]], {x, 0, Infinity}]
```

$$\text{ConditionalExpression}\left[\frac{\pi^2 + 3\,\text{Log}[2]^2 + 6\,\text{PolyLog}\left[2, -\frac{1}{2}\right]}{6\,u}, \text{Re}[u] > 0\right]$$

In[1]:=
```
fb = Log[1 + 2 * Exp[-u * x]] /. {u → 1, x → Infinity};
fa = Log[1 + 2 * Exp[-u * x]] /. x → 0;
fbd = D[Log[1 + 2 * Exp[-u * x]], x] /. {u → 1, x → Infinity};
fad = D[Log[1 + 2 * Exp[-u * x]], x] /. x → 0;
eulmac = (fb + fa) / 2 +
  (fbd - fad) / 12 (* Higher order derivatives can be ignored in this case *)
```

Out[5]= $\frac{u}{18} + \frac{\text{Log}[3]}{2}$

In[6]:=
```
Gapprox = (π^2 + 3 Log[2]^2 + 6 PolyLog[2, -1/2]) / (6 u) + eulmac
```

Out[6]= $\frac{u}{18} + \frac{\text{Log}[3]}{2} + \frac{\pi^2 + 3\,\text{Log}[2]^2 + 6\,\text{PolyLog}\left[2, -\frac{1}{2}\right]}{6\,u}$

```
(* ---------------------------------- *)
(* For saddle point method see e.g. https://
  ac.cs.princeton.edu/online/slides/AC08-Saddle.pdf
    or https://arxiv.org/ftp/arxiv/papers/1508/1508.01796.pdf *)
(* Saddle-point equation x*f'[x]/f[x] = n+1 *)
```

In[7]:=
```
D[Log[f[Exp[-u]]], u]
```

Out[7]= $-\frac{e^{-u} f'\left[e^{-u}\right]}{f\left[e^{-u}\right]}$

In[8]:=
```
Solve[-D[Gapprox, u] == n + 1, u]
```

Out[8]= $\left\{\left\{u \to -\frac{\sqrt{3\left(\pi^2 + 3\,\text{Log}[2]^2 + 6\,\text{PolyLog}\left[2, -\frac{1}{2}\right]\right)}}{\sqrt{19 + 18\,n}}\right\}, \left\{u \to \frac{\sqrt{3\left(\pi^2 + 3\,\text{Log}[2]^2 + 6\,\text{PolyLog}\left[2, -\frac{1}{2}\right]\right)}}{\sqrt{19 + 18\,n}}\right\}\right\}$

In[9]:= uapprox = $\frac{\sqrt{3\left(\pi^2+3\,\text{Log}[2]^2+6\,\text{PolyLog}\left[2,-\frac{1}{2}\right]\right)}}{\sqrt{19+18\,n}}$;

In[10]:= (* Approximation of Log(1/x^n)+ G[u] = n*u + G[u] *)
Series[n * u + Gapprox /. u → uapprox, {n, Infinity, 1}]

Out[10]= $$\sqrt{\frac{2}{3}\left(\pi^2+3\,\text{Log}[2]^2+6\,\text{PolyLog}\left[2,-\frac{1}{2}\right]\right)}\sqrt{n}+\frac{\text{Log}[3]}{2}+\frac{1}{18}\sqrt{\frac{1}{6}\left(\pi^2+3\,\text{Log}[2]^2+6\,\text{PolyLog}\left[2,-\frac{1}{2}\right]\right)}\sqrt{\frac{1}{n}}+O\left[\frac{1}{n}\right]^{3/2}$$

(* Calculating of b[x] = x*f'[x]/f[x] + x^2*G''[u] = n + Exp[-2*u]*G''[u] *)

In[11]:= Series[n + Exp[-2 u] * D[Gapprox, {u, 2}] /. u → uapprox, {n, Infinity, 0}]

Out[11]= $$2\sqrt{\frac{6}{\pi^2+3\,\text{Log}[2]^2+6\,\text{PolyLog}\left[2,-\frac{1}{2}\right]}}\,n^{3/2}-3\,n+\frac{\left(19+4\,\pi^2+12\,\text{Log}[2]^2+24\,\text{PolyLog}\left[2,-\frac{1}{2}\right]\right)\sqrt{n}}{\sqrt{6\left(\pi^2+3\,\text{Log}[2]^2+6\,\text{PolyLog}\left[2,-\frac{1}{2}\right]\right)}}-\frac{2}{9}\left(19+2\,\pi^2+6\,\text{Log}[2]^2+12\,\text{PolyLog}\left[2,-\frac{1}{2}\right]\right)+\sqrt{O\left[\frac{1}{n}\right]}$$

In[12]:= bapprox = $2\sqrt{\frac{6}{\pi^2+3\,\text{Log}[2]^2+6\,\text{PolyLog}\left[2,-\frac{1}{2}\right]}}\,n^{3/2}$;

(* Final asymptotics: a(n) ~ Exp[G[r]] / (Sqrt[2*Pi*b[r]] * r^n) *)
(* Result must be divided by 3, because we also counted the term for k=0 (1+2*x^k), but the original Product started from 1 *)

In[13]:= Exp$\left[\sqrt{\frac{2}{3}\left(\pi^2+3\,\text{Log}[2]^2+6\,\text{PolyLog}\left[2,-\frac{1}{2}\right]\right)}\sqrt{n}+\frac{\text{Log}[3]}{2}\right]$ / Sqrt[2 * Pi * bapprox] / 3

Out[13]= $$\frac{e^{\sqrt{n}\sqrt{\frac{2}{3}\left(\pi^2+3\,\text{Log}[2]^2+6\,\text{PolyLog}\left[2,-\frac{1}{2}\right]\right)}}\left(\frac{1}{2}\left(\pi^2+3\,\text{Log}[2]^2+6\,\text{PolyLog}\left[2,-\frac{1}{2}\right]\right)\right)^{1/4}}{2\times 3^{3/4}\sqrt{n^{3/2}}\sqrt{\pi}}$$

(* Numerical verification, Richardson extrapolation is used *)

In[14]:= << D:\\MathematicaJ\\A032302.TXT

In[15]:= Length[A032302]

Out[15]= 100 000

In[16]:=
```
$MaxExtraPrecision = 1000;
funs[n_] := (A032302[[n]]) /
  (E^(Sqrt[n] Sqrt[2/3 (π^2 + 3 Log[2]^2 + 6 PolyLog[2, -1/2])]) (1/2 (π^2 + 3 Log[2]^2 + 6 PolyLog[2, -1/2]))^(1/4) /
   (2 × 3^(3/4) Sqrt[n^(3/2)] Sqrt[π]));
Do[Print[N[Sum[(-1)^(m + j) * funs[j * Floor[Length[A032302] / m]] *
     j^(m - 1) / (j - 1)! / (m - j)!, {j, 1, m}], 40]], {m, 10, 100, 10}]
```

```
0.9999282551634778296717579824429622853824
0.9999493046215144588180817604208153294847
0.9999586104410910755589983363746427336059
0.9999641589014781946609045042908804345890
0.9999679433690226417945536045934935797305
0.9999707308855628624656117218754751369645
0.9999729022359156502592732855107881200078
0.9999746575358246396537088344491182603457
0.9999761057673633005578612352298208334321
0.9999773331190249221297009628430308388555
```